

The Hierarchy of Factor Invariance

Phil Ender

UCLA IDRE Statistical Consulting Group

2013 Stata Conference New Orleans

July 19, 2013

Apologia

This presentation is in html because it was too difficult to fit all of the CFA output onto traditional slides. And, yes, there is probably too much output shown.

Introduction

Whenever researchers conduct confirmatory factor analysis with multiple groups, the issue of factor or measurement invariance comes up.

At its most basic level, factor invariance is whether the factors in each group are measuring the same thing. There are a number of ways in which invariance can be assessed. The table below presents the hierarchy of factor invariance in ordered by the number of constraints placed on the model, running from the fewest to the most constraints. The constraints are cumulative through the hierarchy.


The Hierarchy of Factor Invariance

- 1 Configurational(Dimensional) Invariance -- equal number of factors
- 2 Metric (Pattern) Invariance -- equal loadings
- 3 Strong (Scalar) Invariance -- equal intercepts
- 4 Strict Invariance -- equal residuals
- 5 Strict Invariance + Equal Factor Means
- 6 Strict Invariance + Equal Factor Means & Factor Variances

The Data

`use http://www.ats.ucla.edu/stat/data/hsbdemo, clear`

The dataset contains four indicator variables (read, write, math & science) that we will use in our single-factor model. The diagram for a single-group model looks like this:



We will use female (0=male, 1=female) as the multiple-group indicator.

Configurational (Dimensional) Invariance

Configurational or dimensional factor invariance implies that the number of factors are the same within each group. One way we can verify this is by running separate PCAs for each group.

```
. pca read write math science if female==0
. pca read write math science if female==1
```

[Edited Output]

| Component | Male Eigenvalues | Female Eigenvalues |
|-----------|---------------------|-----------------------|
| Comp1 | 2.88982 | 2.95214 |
| Comp2 | .409958 | .421286 |
| Comp3 | .375182 | .3516 |
| Comp4 | .325036 | .274972 |

From the eigenvalues it seems rather clear that there is only one factor within each group. This is not a big surprise since each of the indicators are normed test scores in each of four academic areas.

Separate CFAs for each group

Before continuing on with the hierarchy of factor invariance, it is a good idea to run a CFA separately for each group to see if there are any problems estimating the models. The results for this step will be the same as the multiple-group analysis with all parameters allowed to vary freely and, as such, serves as a check that the multiple-group analysis is coded correctly.

For identification purposes **sem** will fix the loading of the first indicator to one. Additionally, we will also set the intercept and the residual variance for the first indicator to zero in to assist in estimating (identifying) the factor mean.

```
. sem (F1 -> read write math science)    ///
```

```

      (read <- F1@1 _cons@0) if female==0, ///
      mean(F1) variance(e.read@0)
. sem (F1 -> read write math science)    ///
      (read <- F1@1 _cons@0) if female==1, ///
      mean(F1) variance(e.read@0)

```

[Edited Output]

| | Males | | Females | |
|-------------|----------|------------------|----------|------------------|
| | Coef. | OIM Std. Err. | Coef. | OIM Std. Err. |
| Measurement | | | | |
| read <- | | | | |
| F1 | 1 | (constrained) | 1 | (constrained) |
| _cons | 0 | (constrained) | 0 | (constrained) |
| write <- | | | | |
| F1 | .6360156 | .0782702 | .5021136 | .0607196 |
| _cons | 16.52388 | 4.214667 | 29.01451 | 3.199544 |
| math <- | | | | |
| F1 | .5597399 | .0765211 | .64698 | .0612728 |
| _cons | 23.37725 | 4.12048 | 18.92367 | 3.228695 |
| science <- | | | | |
| F1 | .6799764 | .0799045 | .5342854 | .0692102 |
| _cons | 17.31158 | 4.30267 | 23.05656 | 3.646948 |
| Mean | | | | |
| F1 | 52.82418 | 1.095334 | 51.73394 | .9589364 |
| Variance | | | | |
| e.read | 0 | (constrained) | 0 | (constrained) |
| e.write | 60.86516 | 9.023252 | 40.28008 | 5.456223 |
| e.math | 58.17521 | 8.624466 | 41.01741 | 5.556098 |
| e.science | 63.43344 | 9.403998 | 52.33273 | 7.088839 |
| F1 | 109.1779 | 16.18561 | 100.2319 | 13.57712 |
| LR chi2(3) | | = 31.05 | 53.48 | |

In general the factor loadings look fairly similar. However, there seems to be variability in some of the intercepts and residual variances.

Model 1: Free All Parameters

We begin by running a two-group model in which all of the parameters are allowed to vary freely. Note the **ginvariant(none)** option. No multiple-group model can fit better than this one. Thus, this model is used to see the effect of constraining the loadings for the metric invariance model.

```

. sem (F1 -> read write math science)    ///
      (F1 -> read@1)                    /// /* set loadings to 1 in both groups */
      (read <- F1 _cons@0),              /// /* set intercept to 0 in both groups */
      group(female) ginvariant(none)    ///
      mean(F1) variance(e.read@0)      /* set residual var to zero in both groups */

```

Endogenous variables

Measurement: read write math science

| | | | | |
|-----------|----------|----------|----------|----------|
| e.write | | | | |
| male | 60.86516 | 9.023252 | 45.51748 | 81.38779 |
| female | 40.28008 | 5.456223 | 30.88795 | 52.5281 |
| e.math | | | | |
| male | 58.17521 | 8.624466 | 43.50583 | 77.79084 |
| female | 41.01738 | 5.556092 | 31.45334 | 53.48958 |
| e.science | | | | |
| male | 63.43344 | 9.403998 | 47.43815 | 84.82205 |
| female | 52.33273 | 7.088839 | 40.13027 | 68.2456 |
| F1 | | | | |
| male | 109.1779 | 16.18561 | 81.64773 | 145.9907 |
| female | 100.232 | 13.57713 | 76.86081 | 130.7096 |

Note: [*] identifies parameter estimates constrained to be equal across groups.

LR test of model vs. saturated: $\chi^2(6) = 84.53$, Prob > $\chi^2 = 0.0000$

Once again, you can visually compare the loadings, intercepts and residual variances between males and females.

Model 2: Metric Invariance (loadings invariant)

We check the metric or pattern invariance by constraining the loadings (path coefficients) to be equal across groups. We use the `ginvariant(mcoef)` option to do this. In the code below, `mcoef` stands for the measurement coefficient, i.e., the factor loading.

```
. sem (F1 -> read write math science)    ///
      (F1 -> read@1)                    /// /* set loading to 1 in both groups */
      (read <- F1 _cons@0),              /// /* set intercept to 0 in both groups */
      group(female)                     /// /* multiple-group analysis */
      ginvariant(mcoef)                  /// /* hold loadings equal */
      mean(F1)                           /// /* estimate factor means */
      variance(e.read@0)                 /* fixed residual at 0 */
```

```
Endogenous variables
Measurement:  read write math science
Exogenous variables
Latent:      F1
Fitting target model:
```

```
Iteration 0:  log likelihood = -25597.924
Iteration 10: log likelihood = -2782.2219
```

```
Structural equation model          Number of obs    =      200
Grouping variable = female         Number of groups =       2
Estimation method = ml
Log likelihood      = -2782.2219
```

```
( 1) [read]0bn.female#c.F1 = 1
( 2) [write]0bn.female#c.F1 - [write]1.female#c.F1 = 0
( 3) [math]0bn.female#c.F1 - [math]1.female#c.F1 = 0
( 4) [science]0bn.female#c.F1 - [science]1.female#c.F1 = 0
( 5) [var(e.read)]0bn.female = 0
( 6) [read]0bn.female = 0
( 7) [read]1.female#c.F1 = 1
( 8) [var(e.read)]1.female = 0
( 9) [read]1.female = 0
```

| | | OIM | | | | |
|-------------|--|----------|---------------|-------|-------|----------------------|
| | | Coef. | Std. Err. | z | P> z | [95% Conf. Interval] |
| ----- | | | | | | |
| Measurement | | | | | | |
| read <- | | | | | | |
| F1 | | | | | | |
| [*] | | 1 | (constrained) | | | |
| _cons | | | | | | |
| [*] | | 0 | (constrained) | | | |
| ----- | | | | | | |
| write <- | | | | | | |
| F1 | | | | | | |
| [*] | | .5522246 | .0485991 | 11.36 | 0.000 | .4569722 .6474769 |
| _cons | | | | | | |
| male | | 20.95007 | 2.695888 | 7.77 | 0.000 | 15.66623 26.23392 |
| female | | 26.42207 | 2.587114 | 10.21 | 0.000 | 21.35142 31.49272 |
| ----- | | | | | | |
| math <- | | | | | | |
| F1 | | | | | | |
| [*] | | .6129485 | .048102 | 12.74 | 0.000 | .5186703 .7072267 |
| _cons | | | | | | |
| male | | 20.56656 | 2.664416 | 7.72 | 0.000 | 15.3444 25.78871 |
| female | | 20.68425 | 2.563209 | 8.07 | 0.000 | 15.66045 25.70805 |
| ----- | | | | | | |
| science <- | | | | | | |
| F1 | | | | | | |
| [*] | | .5965753 | .0530537 | 11.24 | 0.000 | .4925918 .7005587 |
| _cons | | | | | | |
| male | | 21.71717 | 2.925668 | 7.42 | 0.000 | 15.98297 27.45138 |
| female | | 19.83406 | 2.831422 | 7.00 | 0.000 | 14.28457 25.38354 |
| ----- | | | | | | |
| Mean | | | | | | |
| F1 | | | | | | |
| male | | 52.82417 | 1.095333 | 48.23 | 0.000 | 50.67736 54.97099 |
| female | | 51.73396 | .9589334 | 53.95 | 0.000 | 49.85449 53.61344 |
| ----- | | | | | | |
| Variance | | | | | | |
| e.read | | | | | | |
| [*] | | 0 | (constrained) | | | |
| e.write | | | | | | |
| male | | 61.63169 | 9.180054 | | | 46.0275 82.52599 |
| female | | 40.53178 | 5.511979 | | | 31.04841 52.91172 |
| e.math | | | | | | |
| male | | 58.48431 | 8.688283 | | | 43.71062 78.25133 |
| female | | 41.13349 | 5.581478 | | | 31.52786 53.66568 |
| e.science | | | | | | |
| male | | 64.19285 | 9.5655 | | | 47.93442 85.96582 |
| female | | 52.7216 | 7.172172 | | | 40.38242 68.83113 |
| F1 | | | | | | |
| male | | 109.1777 | 16.18554 | | | 81.64762 145.9903 |
| female | | 100.2313 | 13.57695 | | | 76.86044 130.7085 |

Note: [*] identifies parameter estimates constrained to be equal across groups.

LR test of model vs. saturated: chi2(9) = 89.02, Prob > chi2 = 0.0000

We won't compare compare the chi-square and degrees of freedom to the previous model just yet. We will wait until after we run the models for the other forms of factor invariance.

Model 3: Strong Invariance (metric invariance plus equal intercepts)

As we step through each of the levels of invariance, we retain the constraints from the previous model and add an additional constraint. For strong or scalar invariance we add the constraint that the intercepts to are equal across groups. In the code below, **mcons** stands for the measurement constant, i.e., the intercept.

```
. sem (F1 -> read write math science) ///
      (F1 -> read@1)                /// /* set loadings to 1 in both groups */
      (read <- F1 _cons@0),          /// /* set intercept to 0 in both groups */
      group(female)                 ///
      ginvariant(mcoef mcons)        ///
      mean(F1) variance(e.read@0)
```

[Output Redacted]

```
LR test of model vs. saturated: chi2(12) = 118.27, Prob > chi2 = 0.0000
```

Model 4: Strict Invariance (strong invariance plus equal residuals)

For strict invariance, we add the constraint that the residual variances are to be equal across groups. We do not need to use the **ginvariant** option because constraining the residual variances to be equal implies the the loadings and intercepts are also equal.

```
. sem (F1 -> read write math science) ///
      (F1 -> read@1)                /// /* set loadings to 1 in both groups */
      (read <- F1 _cons@0),          /// /* set intercept to 0 in both groups */
      group(female)                 ///
      var(e.read@0) var(e.write@v2)  /// /* set residuals to be equal */
      var(e.math@v3) var(e.science@v4) /// /* set residuals to be equal */
      mean(F1)
```

[Output Redacted]

```
LR test of model vs. saturated: chi2(15) = 128.03, Prob > chi2 = 0.0000
```

Model 5: Strict Invariance Plus Factor Means

Next, we constrain the factor means to be equal across groups in addition to all previous constraints.

```
. sem (F1 -> read write math science) ///
      (F1 -> read@1)                /// /* set loadings to 1 in both groups */
      (read <- F1 _cons@0),          /// /* set intercept to 0 in both groups */
      group(female)                 ///
      var(e.read@0) var(e.write@v2)  ///
      var(e.math@v3) var(e.science@v4) ///
      mean(F1@m1)                   /* set factor means to be equal */
```

[Output Redacted]

```
LR test of model vs. saturated: chi2(16) = 128.59, Prob > chi2 = 0.0000
```

Model 6: Strict Invariance Plus Factor Means & Factor Variances

Add one last constraint to our list: Equal factor variances.

```
. sem (F1 -> read write math science)    ///
      (0: F1 -> read@1)                  /// /* set loading to 1 in group 0 */
      (1: F1 -> read@1)                  /// /* set loading to 1 in group 1 */
      (read <- F1 _cons@0),              /// /* set intercept to 0 in both groups */
      group(female)                      /// /* multiple-group analysis */
      var(e.read@0) var(e.write@v2)      /// /* set residuals to be equal */
      var(e.math@v3) var(e.science@v4)  /// /* set residuals to be equal */
      var(F1@v5)                          /// /* set factor variances to be equal */
      mean(F1@m1)                         /* set factor means to be equal */
```

[Output Redacted]

LR test of model vs. saturated: chi2(17) = 128.77, Prob > chi2 = 0.0000

Summary Table

For the purposes of this presentation, we are just going to look at the chi-square values and degrees of freedom for each model. In actual practice, we would want to also look at measures like RMSEA and CFI.

We start off by comparing the metric invariance model to the one in which all parameters are free to vary. The metric invariance model does not fit significantly worse. However, when we compare the strong invariance with metric invariance, the fit does become significantly worse. Even though we didn't meet the strong factor invariance level, we show how the remaining comparisons are computed.

| model | chi2 | df | ref model | delta chi2 | delta df | P |
|-----------------------------|--------|----|-----------|------------|----------|-------|
| Model 1 free all parameters | 84.53 | 6 | - | | | |
| Model 2 metric (loadings) | 89.02 | 9 | 1 | 4.49 | 3 | .2132 |
| Model 3 strong (intercepts) | 118.27 | 12 | 2 | 29.25 | 3 | .0000 |
| Model 4 strict (residuals) | 128.03 | 15 | 3 | 9.76 | 3 | .0207 |
| Model 5 plus factor means | 128.59 | 16 | 4 | .56 | 1 | .4543 |
| Model 6 plus factor var | 128.77 | 17 | 5 | .18 | 1 | .6714 |

Partial Strong Invariance

We did not achieve full strong factor invariance. However, inspection of the the model with all parameters free to vary suggests that the problem may lie with the intercepts for the variable write. Let's try a model with partial strong invariance, in which, the intercepts for math and science are constrained to be equal but are allowed to vary for write.

```
sem (F1 -> read write math science)    ///
      (F1 -> read@1)                  /// /* set loadings to 1 in both groups */
      (read <- F1 _cons@0)            /// /* set intercept to 0 in both groups */
      (0: write <- F1 _cons@i1)       /// /* allow intercepts for write to differ */
      (1: write <- F1 _cons@i2),      /// /* allow intercepts for write to differ */
      group(female)                  ///
      ginvariant(mcoef mcons)         ///
      variance(e.read@0)              /// /* set residual to 0 */
      mean(F1)                        /* estimate factor means */
```

[Output Redacted]

LR test of model vs. saturated: $\chi^2(11) = 91.99$, Prob > $\chi^2 = 0.0000$

Partial Strict Invariance

If we free the residual variances for write, we get a model for partial strict invariance.

```
sem (F1 -> read write math science) ///
(0: F1 -> read@1) /// /* set loadings to 1 in both groups */
(1: F1 -> read@1) ///
(read <- F1 _cons@0) /// /* set intercept to 0 in both groups */
(0: write <- F1 _cons@i1) /// /* allow intercepts for write to differ */
(1: write <- F1 _cons@i2), /// /* allow intercepts for write to differ */
group(female) ///
ginvariant(mcoef mcons) ///
variance(e.read@0) /// /* set residual to 0 */
var(e.math@v3) var(e.science@v4) ///
mean(F1)
```

[Output Redacted]

LR test of model vs. saturated: $\chi^2(13) = 96.03$, Prob > $\chi^2 = 0.0000$

Now that the intercepts and residual variances for write are allowed to vary across groups, we can run models constrain factor means and factor variances to be equal (code not shown). Here is the summary table for the partial invariance models.

Revised Summary Table for Partial Invariance

| model | chi2 | df | ref model | delta chi2 | delta df | P |
|-----------------------|-------|----|--------------|---------------|-------------|-------|
| 1 free all parameters | 84.53 | 6 | - | | | |
| 2 metric (loadings) | 89.02 | 9 | 1 | 4.49 | 3 | .2132 |
| 3 partial strong | 91.99 | 11 | 2 | 2.97 | 2 | .2265 |
| 4 partial strict | 96.03 | 13 | 3 | 4.04 | 2 | .1327 |
| 5 plus factor means | 96.59 | 14 | 4 | .53 | 1 | .4666 |
| 6 plus factor var | 96.77 | 15 | 5 | .21 | 1 | .6468 |

You will note that model 6 (partial strict plus factor means and variances) does not fit significantly worse than model 1 (all parameters free).